2014-05-23.sagews

May 23, 2014

Contents

1 Math 480b Sage Course

1.1 Today: Groups

1.1.1 May 23, 2014

Screencast: `http://youtu.be/iiWPsS2Q87c`

Plan

- Questions
- Homework: everything due Monday by 6am!!
- Presentations: update about final schedule
- Groups

```
# (aside) John Palmieri's story about SNF of adjacency matrix of a graph\
  ....
```

1.2 Groups in Sage: introduction

- Reference Manual: http://www.sagemath.org/doc/reference/groups/

```
G = groups.permutation.Dihedral(4); G
```

```
Dihedral group of order 8 as a permutation group
```

```
list(G)
```

```
[(), (2,4), (1,2)(3,4), (1,2,3,4), (1,3), (1,3)(2,4), (1,4,3,2), (1,4)(2,3)]
```

```
v = G.gens(); v
```

```
[(1,2,3,4), (1,4)(2,3)]
```

```
a = v[0]; b = v[1]
```

```
a*b
```

```
(1,3)
```

```
len(dir(G))
```

```
349
```

1.3 Disjoint Cycle Notation

- Fact: All finite groups can be viewed as a set of permutations of the integers $\{1, \ldots, n\}$.
- Disjoint cycle notation is the standard notation for describing such a permutation.

The notation $(a_1, a_2, ..., a_k)$ means: the permutation

$$a_1 \mapsto a_2,\ a_2 \mapsto a_3, \ldots, a_{k-1} \mapsto a_k,\ a_k \mapsto a_1$$

```
Exercise:  what does
   (2,3,1,4)
do?
    1 |--> 4
    2 |--> 3
    3 |--> 1
    4 |--> 2
```

Disjoint cycle notation: combine multiple cycles as above, e.g.,
(2,3,1,4)(5,7,6)
means the permutation that maps 2,3,1,4 as above, and does this to 5,6,7:
5 — 7 6 — 5 7 — 6

```
S = SymmetricGroup(7)
a = S([(2,3,1,4), (5,7,6)])
b = S((1,2,5,7))
```

```
a*b
```

```
(1,4,5)(2,3)(6,7)
```

```
b*a
```

```
(1,3)(2,7,4)(5,6)
```

```
"first do b then do a".
1 |-> 3
2 |-> 7
3 |-> 1
4 |-> 2
5 |-> 6
6 |-> 5
7 |-> 4

(1,3)(2,7,4)(5,6)
```

1.4 The Symmetric Group

The group of all permutations of the integers $\{1, 2, \ldots, n\}$ is called the symmetric group S_n.

```
S = SymmetricGroup(4); S
```

```
Symmetric group of order 4! as a permutation group
```

```
S.cardinality()
```

```
24
```

```
factorial(4)
```

```
24
```

```
list(S)  # exercise: find our cycle (2,3,1,4) below...
```

```
[(), (3,4), (2,3), (2,3,4), (2,4,3), (2,4), (1,2), (1,2)(3,4), (1,2,3), (1,2,3,4),
(1,2,4,3), (1,2,4), (1,3,2), (1,3,4,2), (1,3), (1,3,4), (1,3)(2,4), (1,3,2,4), (1,4,3,2),
(1,4,2), (1,4,3), (1,4), (1,4,2,3), (1,4)(2,3)]
```

```
S.gens()
```

```
[(1,2,3,4), (1,2)]
```

```
# how to make an element:
a = S([(1,2),(3,4)]); a
```

```
(1,2)(3,4)
```

```
a(1)
```

```
2
```

```
a(3)
```

```
4
```

```
b = S([2,3,1,4]); b
```

```
(1,2,3)
```

```
# exercise: does this mean "do b then a" or does it mean "do a then b"?
a * b
```

```
(1,3,4)
```

1.5 Subgroups

A subgroup G of S_n is a subset such that:

- if $a, b \in G$ then $ab \in G$.
- for each $a \in G$ there is $b \in G$ such that $ab = ()$, where $()$ means the identity permutation, which leaves everything fixed.

Given any set of elements of S_n, the set of everything you can get by multiplying those elements together is a subgroup, called the subgroup generated by them.

```
a = S([(1,2),(3,4)]); a
b = S([2,3,1,4]); b
```

```
(1,2)(3,4)
(1,2,3)
```

```
G = S.subgroup([a,b]); G
```

```
Subgroup of (Symmetric group of order 4! as a permutation group) generated by [(1,2)(3,4),
(1,2,3)]
```

```
S.cardinality()
```

```
24
```

```
G.cardinality()
```

```
12
```

```
list(G)   # this is called "The alternating group A_4".  Groups often \
   have names.   One of the most
          # important theorems of the 20th century was a complete \
            classification of all finite groups!!
```

```
[(), (2,3,4), (2,4,3), (1,2)(3,4), (1,2,3), (1,2,4), (1,3,2), (1,3,4), (1,3)(2,4),
(1,4,2), (1,4,3), (1,4)(2,3)]
```

1.6 Where do groups come from? Graphs

On Wednesday we saw how subgroups arise naturally as the symmetries (=automorphism groups) of graphs.

```
g = graphs.WheelGraph(6)
g.plot()
```




```
G = g.automorphism_group(); G
```
Permutation Group with generators [(2,5)(3,4), (1,2,3,4,5)]

```
H = groups.permutation.Dihedral(5); H
```
Dihedral group of order 10 as a permutation group

```
G.is_isomorphic(H)
```
True

```
G.cardinality()
```
10

```
list(G)
```

```
[(), (2,5)(3,4), (1,2)(3,5), (1,2,3,4,5), (1,3)(4,5), (1,3,5,2,4), (1,4)(2,3),
(1,4,2,5,3), (1,5,4,3,2), (1,5)(2,4)]
```

1.7 Where do groups come from? Games

```
G = groups.permutation.RubiksCube(); G
```

```
The Rubik's cube group with generators R,L,F,B,U,D in SymmetricGroup(48).
```

```
G.cardinality()
```

```
43252003274489856000
```

```
43252003274489856000.0` / factorial(48)
```

```
3.4841547738970719308e-42
```

```
show(factor(G.cardinality()))
```

$$2^{27} \cdot 3^{14} \cdot 5^{3} \cdot 7^{2} \cdot 11$$

```
G.gens()
```

```
[(1,14,48,27)(2,12,47,29)(3,9,46,32)(33,35,40,38)(34,37,39,36),
(14,22,30,38)(15,23,31,39)(16,24,32,40)(41,43,48,46)(42,45,47,44),
(6,25,43,16)(7,28,42,13)(8,30,41,11)(17,19,24,22)(18,21,23,20),
(1,17,41,40)(4,20,44,37)(6,22,46,35)(9,11,16,14)(10,13,15,12),
(3,38,43,19)(5,36,45,21)(8,33,48,24)(25,27,32,30)(26,29,31,28),
(1,3,8,6)(2,5,7,4)(9,33,25,17)(10,34,26,18)(11,35,27,19)]
```

```
# explain why moves of the cube defines a subgroup of S_{48}....
```

Google Rubiks Cube

1.8 Where do groups come from? matrices over finite fields

- recall finite fields, e.g., $\mathbf{F}_5 = 0, 1, 2, 3, 4$.

```
F5 = GF(5)
F5
```

```
Finite Field of size 5
```

```
list(F5)
```

```
[0, 1, 2, 3, 4]
```

```
a = F5(4); b = F5(2)
print 'a     =', a
print 'b     =', b
print "a + b =", a+b
print "a * b =", a*b
```

```
a     = 4
b     = 2
a + b = 1
```

```
a * b = 3
```

The set of invertible 2×2 matrices with entries in a specific finite field is a group.

```
G = GL(2, F5); G
```

```
General Linear Group of degree 2 over Finite Field of size 5
```

```
G.cardinality()
```

```
480
```

```
show(G.gens())
```

$$\left(\begin{pmatrix} 2 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 4 & 1 \\ 4 & 0 \end{pmatrix} \right)$$

1.8.1 You can get other groups:

- changing 2 to any $n \geq 1$.
- considering only matrices with determinant 1
- Considering only upper triangular invertible matrices.

and many other things.
This leads to a major area of mathematics called representation theory.

1.9 Where do groups come from? a million other surprising places

- http://www.sagemath.org/doc/reference/groups/sage/groups/perm_gps/permgroup_named.html

There are also infinite groups e.g., invertible 2×2 matrices with real number entries.